Gnuplot: Exercícios do livro do Thomas - 2

Ulysses Sodré e Sônia F. L. Toffoli

Londrina-PR, 27 de Junho de 2007, arquivo: sacgnu02.tex

1 Mudando a lista de configuração

1. Após plotar a função $f(x) = x * \sin(x)$, será criada outra janela com o gráfico pedido.
2. Redimensionamos a janela gráfica, clicando com o **botão direito** do mouse sobre a janela gráfica, para ver uma lista como [fig.1]



Figura 1: Detalhes da janela gráfica

3. Detalhes sobre a lista.
   (a) **Bring to top**: Mantém o programa visualmente na tela.
   (b) **Color**: mostra gráficos coloridos. Se **Color** estiver desmarcado, os gráficos ficarão em branco e preto.
   (c) **Copy to Clipboard**: Guarda na memória do PC o gráfico para inserção em outro programa gráfico.
   (d) **Background**: Muda a cor de fundo dessa janela.

(e) **Choose font**: Altera a fonte e tamanho de escrita da legenda e das escalas.

(f) **Line Styles**: Oferece detalhes como: espessura, cor, tipo de tracejado, etc., sobre as linhas.

(g) **Print**: Imprime o gráfico.

(h) **Update wgnuplot.ini**: Atualiza as suas mudanças e mantém esta estrutura salva a partir desse momento.

2 Alterando a posição da legenda

1. A legenda no Canto superior direito pode ser vista com:

```
set key right top title 'Legenda'
plot sin(x)
```



Figura 2: Legenda à direita em cima

2. Outras formas alternativas para legendas são:

```
set key right bottom title 'Leg' spacing 3
set key right bottom title 'Leg' box 3 spacing 3
set key below Left title 'Leg' box 3 spacing 3
set key outside Left title 'Leg' box 3 spacing 3
set key top outside Left title 'Leg' box 3 spacing 3
```

3 Código muito longo em uma linha

Se a linha tem o código **muito longo**, podemos *quebrá-la* em várias linhas com um espaço e uma barra invertida para continuar a digitar o código na linha de baixo.

O código

```
set key top outside Left title 'Legenda' box 3 spacing 3
```

também pode ser digitado como:

```
set key top outside Left title 'Legenda' \
box 3 spacing 3
```

ou como

```
set key top outside \
Left title 'Legenda' \
box 3 spacing 3
```

4 Função definida por partes

Uma função definida por partes tem a forma

$$f(x) = \begin{cases} \sin(x) & \text{se} \quad x < 0 \\ 5 - x & \text{se} \quad x > 0 \end{cases}$$

onde cada parte é associada a um intervalo. A notação `1/0` usada abaixo representa o **infinito** no Gnuplot.

```
set samples 300
f(x) = x<0 ? sin(x) : x>0 ? 5-x : 1/0
plot f(x)
```



Figura 3: Função definida em duas partes

5 Domínio dado e contradomínio automático

Plotaremos agora funções em um domínio dado com a escala y automática, isto é, o contradomínio de f coincide com a imagem de f.

Plotaremos $f(x) = 2\sin(3x)$ sobre $[-2, 2]$, $g(x) = 2\sin(x)$ sobre $[-\pi, \pi]$ e $h(x) = \sin(x)$ sobre $[-\pi, \pi]$ com a escala y automática.

```
set autoscale y
plot [-2:2] 2*sin(3*x)
plot [-pi:pi] 2*sin(x)
```



Figura 4: domínio fixo e contradomínio automático

Não é necessário mudar o contradomínio para as próximas funções que serão plotadas, pois ele já está no automático com **autoscale y**.

6 Domínio automático e contradomínio fixado

Plotamos várias funções em um domínio automático e contradomínio fixado, *desligando* (se estiver *ligada*) a escala automática para y e *ligando* a escala automática para x.

Plotamos $f(x) = 2\sin(3x)$ com o contradomínio $[-4, 4]$ e $g(x) = 2\sin(3x)$ com o contradomínio $[-\pi, \pi]$ com a escala x automática, com:

```
set noautoscale y
set autoscale x
plot [][-4:4] 2*sin(3*x)
plot [][-pi:pi] 2*sin(3*x)
```

Não é necessário mudar o domínio para as próximas funções, pois este já está no automático com **autoscale x**.

7 Domínio e contradomínio automáticos de novo

Podemos plotar uma ou mais funções com o domínio e o contradomínio automáticos. A função $f(x) = 1/(1+x^2)$ pode ser plotada de uma forma automática, com:

```
set autoscale
plot 1/(1+x**2)
```



Figura 5: Função com escalas automáticas

O comando **set autoscale** volta a exibir o padrão do Gnuplot para gráficos no plano.

8 Relação que não é função

Para plotar a relação $x = y^2$ de $x = 0$ a $x = 25$, [fig.6] com a imagem em $[-5, 5]$, digite:



Figura 6: Gráfico de relação que não é função

```
set key Left bottom left spacing 2
set autoscale y
set xrange [0:25]
plot sqrt(x), -sqrt(x)
```

Quando é possível explicitar $y = f(x)$, plotamos da mesma forma outras relações que não são funções.

9 Gráfico da circunferência

Para plotar o gráfico da circunferência $x^2 + y^2 = 9$, escreva as funções $y = +\sqrt{9 - x^2}$ e $y = -\sqrt{9 - x^2}$, digitando:

```
plot [-3:3] [-3:3] sqrt(9-x**2),-sqrt(9-x**2)
```

Limpe a *memória* (do PC) com **reset**.

10 Curvas paramétricas no plano

Curvas paramétricas no plano, podem ser plotadas com **set parametric** e a variável *muda* t.

1. A circunferência parametrizada por $r(t) = (\cos(t), \sin(t))$ com t real, pode ser plotada com

   ```
   set parametric
   set title "circunferência"
   plot cos(t),sin(t)
   ```

2. Outra circunferência parametrizada $r(t) = (\sin(t), \cos(t))$ [fig.7] pode ser plotada com:

   ```
   plot sin(t),cos(t)
   ```



Figura 7: Curva $r(t) = (\sin(t), \cos(t))$

3. O gráfico da função **sinc()** parametrizada por $r(t) = (t, \sin(t)/t)$ [fig.8] pode ser gerado pelo código:

```
set key box 4 below title 'sinc()'
plot [-12:12] t,sin(t)/t
```



Figura 8: Curva $x(t) = t$ e $y(t) = sen(t)/t$

4. Outras curvas que podem ser plotadas por você, são: $r(t) = (t^2, t^3)$, $r(t) = (t, \tan t)$, $r(t) = (2\cos t, t)$ e $r(t) = (\sin(t), 2t)$.

11 Várias curvas parametrizadas no mesmo gráfico

Podemos plotar várias seções cônicas parametrizadas em um mesmo gráfico, após definir o domínio das variáveis x e y. Cada **par** de funções representa uma curva parametrizada no plano R^2.[fig.9]



Figura 9: Cônicas (curvas parametrizadas)

```
set xrange [-3:3]
set yrange [-3:3]
set title "Cônicas: formas paramétricas"
plot 3*cos(t),2*sin(t), 2*cos(t),3*sin(t)
```

Podemos eliminar o título com `set notitle` na versão 3.7.3. ou com o código `unset title` na versão 4.0.

As curvas $r(t) = (\tan(t), t)$ e $s(t) = (t, \tan(t))$ podem ser plotadas no mesmo gráfico com:

```
set xrange [-5:5]
set yrange [-5:5]
plot tan(t),t, t,tan(t)
```

12 Funções não definidas em zero

Para traçar funções não definidas em zero, devemos definir o parâmetro muito próximo de zero.

1. Plotamos $r(t) = (t, \log(t))$ e $s(t) = (-t, \log(t))$ tomando o domínio em $[0.01, 3]$, pois t pode ser não nulo.

   ```
   set trange [0.01:3]
   plot t,log(t), -t,log(t)
   ```

2. Acrescentamos mais 2 curvas ao gráfico anterior, com:

   ```
   plot t,log(t), \
   -t,log(t), \
   sin(t), t**2, \
   -sin(t),t**2
   ```



Figura 10: Curvas não definidas em zero

13 Curvas em coordenadas polares

Limpe a *memória* (do PC) com **reset**.

Plotaremos curvas em coordenadas polares com **set polar**, mas primeiro definiremos o domínio.

Três circunferências centradas na origem com raios 0.5, 1 e 1.5, com:

```
set polar
set zeroaxis
set title 'Três circunferências'
plot .5, 1, 1.5
```



Figura 11: Três Circunferências

As curvas $r = 2\sqrt{\cos(t)}$ e $r = -2\sqrt{\cos(t)}$, podem ser plotadas com:

```
plot 2*sqrt(cos(t)),-2*sqrt(cos(t))
```



Figura 12: $r = 2\sqrt{\cos(t)}$ e $r = -2\sqrt{\cos(t)}$

Os gráficos das equações $r = \sin(4t)$ e $r = \cos(4t)$ são plotadas com:

```
plot sin(4*t),cos(4*t)
```

14 Curvas em coordenadas polares em um retângulo

A cardióide $r = 1 - \sin(t)$ no retângulo $[-2, 2] \times [-2, 0.5]$, pode ser plotada com:

```
set xrange [-2:2]
set yrange [-2:0.5]
plot 1-sin(t)
```



Figura 13: Uma cardióide

15 Borboleta

A *borboleta* abaixo pode ser plotada com o código:



Figura 14: Borboleta

```
set autoscale xy
borboleta(t)=exp(cos(t))-2*cos(4*t)+sin(t/12)**5
set samples 800
set title "Borboleta"
set nokey
plot borboleta(t)
```

16 Exercícios do livro do Thomas - 2

1. Pag.125, Exemplo 1: Plotar $r(t) = (t\cos(t), t\sin(t))$. O código é:
```
set parametric
set zeroaxis
n=2
plot [0:n*pi] t*cos(t), t*sin(t)
n=3
rep
```
   Use outros valores pares de n.

2. Pag.126, Exemplo 3: Plotar $r(t) = (1/t, \sin(t))$ com $t > 0$.
   Código: `plot [0.01:2] 1/t, sin(t)`

3. Pag.130, Exemplo 6: Plotar $r(t) = (3\cos(t), 3\sin(t))$ em $[0, 2\pi]$.
```
plot [0:2*pi] 3*cos(t), 3*sin(t)
```

4. Pag.133 - Plotar cada função indicada:

   02 $r(t) = (2\cos(2t), 2\sin(t))$

   03 $r(t) = \pm(1/\cos(t), \tan(t))$ em $[-\pi/3, \pi/3]$.

   04 $r(t) = (2\log(t+1), t^2)$ em $[-0.99, 5]$.

   23 $r(t) = (t - \sin(t), 1 - \cos(t))$ em $[-10 : 10]$.

   23 $r(t) = (1 - \cos(t), t - \sin(t))$ em $[-10 : 10]$.

   24 $r(t) = (\exp(4t)/4 - t, \exp(2t))$ em $[0, 2]$

5. Pag.134, Ex.32: Plotar as curvas $r_1(t) = (t-3, (t-3)^2)$ e $r_2(t) = (3t/2 - 4, 3t/2 - 2)$ e obter o valor de t em que $r_1(t) = r_2(t)$.

Os próximos exercícios tratam sobre gráficos de **curvas polares**.

Limpe a *memória* do computador com **reset**.

1. Pag.149, Exemplo 1: Plotar as curvas polares $r = 1$ e $r = -1$.
   Código: `plot 1, -1`

2. Pag.149, Exemplo 2: Plotar o conjunto de pontos tal que:

   (1) $1 \le r \le 2; \qquad 0 \le t \le \pi/2$

(?) $-3 \le r \le 2; \qquad t = \pi/4$

(?) $r \le 0; \qquad t = \pi/4$

(?) $2\pi/3 \le t \le 5\pi/6$

3. Pag.155, Ex.76: Plotar as rosáceas $r = 2\sin(nt)$ para $n = -4$, $n = -3$, $n = -2$, $n = -1$, $n = 1$, $n = 2$, $n = 3$, $n = 4$. Para cada n, qual é o domínio para que a curva volte ao local de origem?

```
set polar
r(t)=2*sin(n*t)
plot [-pi:pi] r(t), n=-4
```

Mude a última linha do código para ver as outras curvas.

4. Pag.155, Ex.77: Plotar a curva $r = 1 - 2\sin(3t)$.

5. Pag.160, Exemplo 4: Plotar a Limaçon de Pascal: $r = 1 + 2\cos(t)$.

6. Pag.163: Plotar as curvas polares.

03 $r = 2 - 3\sin(t)$

04 $r = 3(1 - \cos(t))$

06 $r = 2\cos(3t)$

14 $r = a(1 + \cos(t))$

17 $r^2 = a^2 \sin(2t)$ (em duas curvas)

18 $r^2 = a\sin(3t)$ (em duas curvas)

7. Pag.163: Plotar os pares de curvas polares.

Pg.154 $r = \cos(2t)$ e $r = \sin(2t)$.

19 $r = 2\cos(t)$ e $r = 2\sin(t)$

20 $r = 1$ e $r = 2\sin(t)$

21 $r = 2$ e $r = 2(1 - \cos(t))$

22 $r = 2(1 + \cos(t))$ e $r = 2(1 - \cos(t))$

23 $r^2 = 6\cos(2t)$ e $r = \sqrt{3}$

24 $r = 3\cos(t)$ e $r = 1 + \cos(t)$

25 $r = -2\cos(t)$ e $r = 1$

8. Pag.149, Exemplos 3 e 9: Plotar primeiramente as curvas polares $r^2 = 4\cos(t)$ e depois acrescentar a curva $r = 1 - \cos(t)$.